In[1]:=
```
(* General setup *)
ClearAll["Global`*"]
$Assumptions =
  rho > 0 && rhoe > 1 / 2 * rhov ^ 2 / rho && gamma > 1 && (alpha > 0 || alpha < -gamma);

(* conserved variables *)
u = {rho, rhov, rhoe};

(* primitive variables *)
v = rhov / rho;
p = (gamma - 1) * (rhoe - 1 / 2 * rho * v ^ 2);

flux = {rho * v, rho * v ^ 2 + p, (rhoe + p) * v};

(* entropy and entropy flux *)
entropy = -(gamma + alpha) / (gamma - 1) * rho * (p / rho ^ gamma) ^ (1 / (alpha + gamma));
entropyflux = entropy * v;

(* entropy variables *)
w =
  rho / p * (p / rho ^ gamma) ^ (1 / (alpha + gamma)) * {-alpha / (gamma - 1) * p / rho - 1 / 2 * v ^ 2, v, -1};

(* flux potential *)
psi = rhov * (p / rho ^ gamma) ^ (1 / (alpha + gamma));
```

In[11]:=
```
(* Verify expression of the entropy variables and the flux potential *)

FullSimplify[D[entropy, {u[[1]]}] == w[[1]]]
FullSimplify[D[entropy, {u[[2]]}] == w[[2]]]
FullSimplify[D[entropy, {u[[3]]}] == w[[3]]]

FullSimplify[w.flux - entropyflux == psi]
```

Out[11]= True

Out[12]= True

Out[13]= True

Out[14]= True

In[15]:=
```
(* EC and PEP density flux for constant pressure and velocity *)
wL = Simplify[
    w /. {rhoe → pLR / (gamma - 1) + 1 / 2 * rhov ^ 2 / rho} /. {rhov → rho * vLR} /. {rho → rhoL}];
psiL = Simplify[psi /. {rhoe → pLR / (gamma - 1) + 1 / 2 * rhov ^ 2 / rho} /.
      {rhov → rho * vLR} /. {rho → rhoL}];

wR = Simplify[
    w /. {rhoe → pLR / (gamma - 1) + 1 / 2 * rhov ^ 2 / rho} /. {rhov → rho * vLR} /. {rho → rhoR}];
psiR = Simplify[psi /. {rhoe → pLR / (gamma - 1) + 1 / 2 * rhov ^ 2 / rho} /.
      {rhov → rho * vLR} /. {rho → rhoR}];

$Assumptions = $Assumptions && pLR > 0 && rhoL > 0 && rhoR > 0;

fnumrhov = fnumrho * vLR + pLR;
fnumrhoe = 1 / 2 * fnumrho * vLR ^ 2 + gamma / (gamma - 1) * pLR * vLR;
fnum = {fnumrho , fnumrhov , fnumrhoe};

solFnumrho = FullSimplify @ Solve[(wR - wL).fnum - (psiR - psiL) == 0, fnumrho]

FullSimplify [(fnumrho /. solFnumrho [[1]]) ==
  - gamma / alpha * (rhoR ^ (alpha / (alpha + gamma)) - rhoL ^ (alpha / (alpha + gamma))) /
    (rhoR ^ (- gamma / (alpha + gamma)) - rhoL ^ (- gamma / (alpha + gamma))) * vLR]
```

Out[23]=
$$\left\{\left\{\text{fnumrho} \to \frac{\text{gamma}\left(-\text{rhoL}^{\frac{\text{gamma}}{\text{alpha}+\text{gamma}}}\ \text{rhoR} + \text{rhoL}\ \text{rhoR}^{\frac{\text{gamma}}{\text{alpha}+\text{gamma}}}\right)\text{vLR}}{\text{alpha}\left(\text{rhoL}^{\frac{\text{gamma}}{\text{alpha}+\text{gamma}}} - \text{rhoR}^{\frac{\text{gamma}}{\text{alpha}+\text{gamma}}}\right)}\right\}\right\}$$

Out[24]= True

In[25]:=
```
(* EC density flux for velocity and special pressure ratio *)
pL = pR * (rhoL / rhoR) ^ (alpha / (alpha + gamma - 1));

wL = Simplify[
    w /. {rhoe → pL / (gamma - 1) + 1 / 2 * rhov ^ 2 / rho} /. {rhov → rho * vLR} /. {rho → rhoL}];
psiL = Simplify[psi /. {rhoe → pL / (gamma - 1) + 1 / 2 * rhov ^ 2 / rho} /.
      {rhov → rho * vLR} /. {rho → rhoL}];

wR = Simplify[
    w /. {rhoe → pR / (gamma - 1) + 1 / 2 * rhov ^ 2 / rho} /. {rhov → rho * vLR} /. {rho → rhoR}];
psiR = Simplify[psi /. {rhoe → pR / (gamma - 1) + 1 / 2 * rhov ^ 2 / rho} /.
      {rhov → rho * vLR} /. {rho → rhoR}];

$Assumptions = $Assumptions && pR > 0 && rhoL > 0 && rhoR > 0;

Clear[fnumrhov]
Clear[fnumrhoe]
fnum = {fnumrho, fnumrhov, fnumrhoe};

solFnumrho = FullSimplify @ Solve[PowerExpand[(wR - wL).fnum - (psiR - psiL)] == 0, fnumrho]

FullSimplify[(fnumrho /. solFnumrho[[1]]) ==
  -(gamma - 1) / alpha * (rhoR ^ (alpha / (alpha + gamma - 1)) - rhoL ^ (alpha / (alpha + gamma - 1))) /
    (rhoR ^ ((1 - gamma) / (alpha + gamma - 1)) - rhoL ^ ((1 - gamma) / (alpha + gamma - 1))) * vLR]
```

Out[34]=
$$\left\{\left\{\text{fnumrho} \to \left((-1+\text{gamma})\,\text{rhoR}^{\frac{\text{alpha}}{\text{alpha}+\text{gamma}}}\left(\text{rhoL}^{1+\frac{\text{alpha}}{(-1+\text{alpha}+\text{gamma})(\text{alpha}+\text{gamma})}}\,\text{rhoR}^{\frac{\text{gamma}}{\text{alpha}+\text{gamma}}} - \text{rhoL}^{\frac{\text{gamma}}{\text{alpha}+\text{gamma}}}\,\text{rhoR}^{1+\frac{\text{alpha}}{(-1+\text{alpha}+\text{gamma})(\text{alpha}+\text{gamma})}}\right)\text{vLR}\right)\Big/\left(\text{alpha}\left(-\text{rhoL}^{\frac{\text{alpha}}{(-1+\text{alpha}+\text{gamma})(\text{alpha}+\text{gamma})}}\,\text{rhoR} + \text{rhoL}^{\frac{\text{gamma}}{\text{alpha}+\text{gamma}}}\,\text{rhoR}^{\frac{\text{alpha}}{-1+\text{alpha}+\text{gamma}}}\right)\right)\right\}\right\}$$

Out[35]= True

In[36]:=
```
(* Show that the expressions of the density flux from above are not identical *)
ClearAll["Global`*"]
rhoL = 1;
rhoR = 2;
gamma = 7 / 5;

difference = (- gamma / alpha * (rhoR ^ (alpha / (alpha + gamma)) - rhoL ^ (alpha / (alpha + gamma))) /
        (rhoR ^ (- gamma / (alpha + gamma)) - rhoL ^ (- gamma / (alpha + gamma)))) -
    (- (gamma - 1) / alpha * (rhoR ^ (alpha / (alpha + gamma - 1)) - rhoL ^ (alpha / (alpha + gamma - 1))) /
        (rhoR ^ ((1 - gamma) / (alpha + gamma - 1)) - rhoL ^ ((1 - gamma) / (alpha + gamma - 1))));

Solve[difference == 0 && alpha > 0, alpha]
Solve[difference == 0 && alpha < - gamma, alpha]
```

Out[41]= {}

Out[42]= {}